
Musique Maestro ! (4)

La ROM version « S » comme « Stéréophonique »

Par Dominique Pessan, André Chéramy et Claude Sittler

La dernière fois, nous avons défini le cahier des charges pour deux versions de la ROM MAESTRO, une version « Stéréophonique » et une version « Monolithique ». Nous allons donc maintenant préparer la ROM « Stéréophonique », incluant les commandes SOUND, MUSIC et PLAY,

les routines de passage d'un PSG à l'autre et la protection de PB5. La prochaine fois, nous décrirons quelques « démos » pour tester cette ROM « MAESTRO-S » et plus tard nous développerons la version « Monolithique ».

LA ROM DE DEPART

Il existe deux variantes de la ROM V1.1, elles ont été décrites dans le CEO-MAG n°75-76 de juillet / août 1996, l'une est fournie avec Euphoric, sous divers noms selon les versions (par exemple : ORIC.ROM ou BASIC11B.ROM), l'autre est distribuée avec la version 1.006 de Sedoric.

Toutes deux ont la même checksum : #1A25. Mais à l'adresse \$F892, on trouve la valeur #58 pour la première version et la valeur #D8 pour la seconde version. C'est de la première

version (BASIC11B.ROM) que nous partirons pour installer nos nouvelles commandes PLAY, MUSIC et SOUND.

Rappelons que les checksums indiquées dans cet article sont en hexadécimal 16 bits et peuvent être obtenues, soit avec la commande CHKSUM de Sedoric V3.0, soit avec l'utilitaire CHKSUM.EXE de Robert Chéramy, soit avec la fonction checksum de la plupart des bons éditeurs hexadécimaux pour PC.

PHILOSOPHIE DE L'IMPLANTATION DES COMMANDES ETENDUES

La ROM étant pleine comme un œuf, nouvelle commande ou commande étendue rime presque toujours avec suppression de quelque chose, à moins que le nouveau code ne soit beaucoup plus optimisé que l'ancien. C'est le cas de la commande SOUND dont le code d'origine est un modèle de travail « bâclé » (comme le signale Fabrice Broche dans « L'Oric à nu ») et que Dominique a optimisé avec brio.

En fait Dominique a pondu deux nouvelles

versions. Dans la première version, il a utilisé un tableau indexé contenant les numéros de registre où écrire les paramètres H (hauteur) et V (volume) en fonction de K (canal). De la gestion de 3 canaux en 144 octets, il est passé à la gestion de 6 canaux en 98 octets.

Soit un gain de 46 octets qui permettait de loger les octets supplémentaires de la nouvelle routine MUSIC mais malheureusement pas de « caser » les octets supplémentaires de PLAY. Qu'à cela ne

tienne, Dominique a encore condensé son code en recalculant le tableau. C'est cette super version que nous vous proposons ci-après.

Nous avons toutefois été amenés à squatter de la place de plusieurs manières :

1) En réduisant la taille du message de Copyright, ce qui nous a permis d'insérer

un greffon destiné à améliorer la gestion de PB5.

- 2) En supprimant la commande EDIT, que personne n'utilise, pour y mettre nos routine de changement de PSG.
- 3) En récupérant la place inutilement occupée par la soi-disant routine « Arrondir le quotient » (cf. Fabrice Broche), sise de \$EFRFA à \$F015, pour y caser le reste de nos petits ajouts.

SOUND K, H, V (Kanal de 1 à 12, Hauteur, Volume)

Cette routine est normalement située de \$FB40 à \$FBCF dans la ROM. Rappelons que pour la compatibilité avec les anciens programmes, les valeurs 4, 5, 6 du paramètre K ont été conservées pour mixer du bruit dans les canaux 1, 2 3. Par analogie, les valeurs 10, 11, 12 ont été ajoutées pour mixer

du bruit dans les 3 nouveaux canaux 7, 8, 9).

La notion de tableau indexé (voir ci-dessous) a été abandonnée au profit du calcul du numéro de registre à adresser en fonction de la valeur se trouvant en \$02E1 (paramètre K, numéro de canal).

Aux valeurs de 02E1 suivantes :												
02E1	1	2	3	4	5	6	7	8	9	10	11	12
correspondent les numéros des registres du PSG où il faudra placer la valeur des paramètres de hauteur H (LL et HH = respectivement octets de poids faible et de poids fort) et volume V:												
LL hauteur	00	02	04	06	06	06	00	02	04	06	06	06
HH hauteur	01	03	05	00	00	00	01	03	05	00	00	00
volume	08	09	0A	08	09	0A	08	09	0A	08	09	0A

Cette table de numéros de registres occupait 36 octets à elle seule.

Or on constate que, évidemment, elle est symétrique. C'est normal puisque le PSG0 a les mêmes numéros de registres que le PGS1. Donc, si on soustrait 6 à toute valeur de \$02E1 supérieure à ou égale à 7, on pointera toujours sur le bon registre.

Occupons nous de la première ligne (registres de la partie basse de la hauteur canaux 1, 2, 3 et registre bruit). Deuxième constatation : un seul registre pour le bruit, le six ! Pourquoi alors le mettre dans une table ? Il suffit de savoir, qu'après avoir éventuellement soustrait 6, si le contenu de \$02E1 est strictement supérieur à 3 alors le registre est

six.

Troisième constatation : Si le contenu de \$02E1 est compris entre 1 et 3 on constate que le registre à adresser a pour numéro : $((\$02E1) - 1) \times 2$ ce qui est obtenu par un SBC #01 suivi d'un ASL. On peut donc supprimer ces valeurs de la table.

Voyons la deuxième ligne (registres de la partie haute de la hauteur canaux, pas de partie haute pour le bruit). Pour \$02E1 compris entre 1 et trois, il suffit de rajouter 1 au calcul de la ligne précédente. Pour \$02E1 compris entre 4 et 6, il s'agit des canaux bruit et les zéros ne signifient rien car il n'y a pas de deuxième registre pour la fréquence du bruit. On peu donc les éliminer aussi.

Pour les valeurs de \$02E1 supérieures à 6, les paramètres sont traités, comme pour la ligne précédente, en soustrayant 6 pour se ramener aux cas précédents.

Il reste la troisième ligne (registre du volume pour les canaux 1, 2, 3). La encore, après s'être ramené au cas où \$02E1 est compris entre 1 et 3, il suffit d'ajouter 7 pour obtenir le numéro du registre cherché. La dernière ligne de la table est ainsi éliminée !

Notons que, comme dans la commande

FB40	20 9C C6	JSR \$C69C	Force Pb5 à 0 (adresse le PSG0 « supplémentaire »)
FB43	AD E1 02	LDA \$02E1	Charge A avec le numéro de canal demandé
FB46	C9 07	CMP #\$07	Teste pour savoir si canal <7
FB48	90 07	BCC \$FB51	Si canal <7, PSG0 OK, aller directement en \$FB51 sans changer de PSG
FB4A	E9 06	SBC #\$06	Sinon PSG1 « normal » requis, soustrait 6 au n° de canal, qui passe donc de 1 à 6
FB4C	48	PHA	Sauve n° canal calculé sur la pile
FB4D	20 95 C6	JSR \$C695	Force Pb5 à 1 (adresse le PSG1 « normal »)
FB50	68	PLA	Récupère le n° de canal corrigé
FB51	C9 04	CMP #\$04	Test pour savoir si canal <4, c'est à dire timbre
FB53	90 0E	BCC \$FB63	Si le canal <4 aller en \$FB63 (timbre)
FB55	E9 03	SBC #\$03	Sinon (BRUIT), soustrait 3 à cette valeur qui passe donc de 1 à 3
FB57	48	PHA	Sauve la nouvelle valeur sur la pile
FB58	A9 06	LDA #\$06	n° registre « hauteur bruit mixable »
FB5A	AE E3 02	LDX \$02E3	Charge dans X la valeur (hauteur) à écrire dans ce registre
FB5D	20 90 F5	JSR \$F590	Ecriture registre PSG
FB60	4C 79 FB	JMP \$FB79	Comme on vient de traiter un bruit mixé, on saute directement au Volume
FB63	48	PHA	Partie traitant le TIMBRE : on sauve le n° de canal
FB64	38	SEC	On a ici le canal entre 1 et 3, on prépare
FB65	E9 01	SBC #\$01	la soustraction puis la multiplication par deux
FB67	0A	ASL	pour calculer le n° de registre « partie basse hauteur timbre »
FB68	48	PHA	On copie ce n° sur la pile
FB69	AE E3 02	LDX \$02E3	Charge dans X la valeur (hauteur) à écrire dans ce registre
FB6C	20 90 F5	JSR \$F590	Ecriture registre PSG
FB6F	68	PLA	On récupère ce n° de registre
FB70	18	CLC	On prépare l'addition
FB71	69 01	ADC #\$01	Calcul du registre suivant (« partie haute hauteur timbre »)
FB73	AE E4 02	LDX \$02E4	Charge dans X la valeur (hauteur) à écrire dans ce registre
FB76	20 90 F5	JSR \$F590	Ecriture registre PSG
FB79	AD E5 02	LDA \$02E5	Place le VOLUME dans l'accumulateur
FB7C	29 0F	AND #\$0F	Ramène sa valeur entre 0 et 15
FB7E	D0 02	BNE \$FB82	Si <>0 on saute
FB80	A9 10	LDA #\$10	Si = 0 (silence), alors le volume prend la valeur 16
FB82	AA	TAX	Volume dans le registre X
FB83	68	PLA	Récupère le n° du canal de 1 à 3 sauvé soit en \$FB57 soit en \$FB63
FB84	18	CLC	prépare l'addition
FB85	69 07	ADC #\$07	Calcule le n° du registre de volume correspondant
FB87	20 90 F5	JSR \$F590	Ecriture registre PSG
FB8A	60	RTS	Fin de la routine

De \$FB8B à \$FBCF il reste 69 octets gagnés pour installer les autres modifications.

SOUND d'origine, la validité des paramètres n'est pas vérifiée (amélioration à faire).

Il faut donc se rappeler qu'il faut notamment avoir $1 \leq K \leq 12$ (Canaux 1-3 timbre et 4-6 bruit sur PSG1 « normal » ; 7-9 timbre et 10-12 bruit sur PSG0 « supplémentaire »).

La commande SOUND, anciennement de \$FB40 à \$FBCF se limite maintenant de \$FB40 à \$FB8A. Voici le listing commenté de la nouvelle commande SOUND :

MUSIC C, O, N, V (Canal de 1 à 6, Octave, Note, Volume)

Le code de la commande MUSIC, qui va de \$FC18 à \$FC5D (plus 2 tables de FC5E à FC77) sera légèrement adapté.

Pour comprendre la commande MUSIC, il faut savoir qu'elle adapte ses propres paramètres selon la syntaxe de SOUND et passe ensuite la main à cette dernière commande pour générer le son.

Nous avons légèrement vampirisé le début de la commande pour insérer un détournement vers un greffon que nous avons placé en \$FB8B, dans la place libérée ci-dessus par la réécriture de la commande SOUND. Pour ceux qui n'ont pas « L'Oric à nu » de Fabrice Broche, nous rappelons ci-dessous le début normal de cette commande.

FC18	A2 E1	LDX #\$E1	Début de la commande MUSIC normale
FC1A	A9 07	LDA #\$04	Vérification que $1 \leq C \leq 3$
FC1C	20 E4 F2	JSR \$F2E4	
FC1F	B0 39	BCS \$FC5A	Traitement de l'erreur
FC21	A2 E3	LDX #\$E3	Vérification que $0 \leq O \leq 7$ etc.

Les instructions écrasées seront évidemment reprises dans notre greffon. Voici notre détournement.

FC18	A2 E1	LDX #\$E1	Charge X avec le numéro de canal demandé
FC1A	A9 07	LDA #\$07	Vérifie que $1 \leq C \leq 6$
FC1C	20 8B FB	JSR \$FB8B	Appelle notre greffon inséré pour traiter les paramètres étendus
FC1F	EA	NOP	La reprise du cours normal de la commande MUSIC se fera donc en \$FC1F
FC20	EA	NOP	Ces deux NOP remplacent le BCS qui n'a plus lieu d'être ici

Et voici notre greffon « MUSIC », situé donc dans la fin de l'ancienne commande SOUND :

FB8B	20 E4 F2	JSR \$F2E4	teste si paramètre <7 et non nul
FB8E	B0 13	BCS \$FBA3	Si non, erreur...
FB90	AD E1 02	LDA \$02E1	Si oui, prend le paramètre dans A pour voir quel PSG est concerné
FB93	29 04	AND #\$04	Teste le bit 2
FB95	F0 11	BEQ \$FBA8	si le b2=0: cas ordinaire (canal < 4) donc on retourne à la commande MUSIC
FB97	AD E1 02	LDA \$02E1	si le b2=1 (canal > 3), on recharge l'accumulateur avec le n° du canal
FB9A	29 FB	AND #\$FB	on force le bit 2 à 0, la valeur résultante passe de 0 à 2
FB9C	18	CLC	Prépare une addition pour ajouter 7, la valeur résultante passe de 7 à 9
FB9D	69 07	ADC #\$07	En effet, pour SOUND les nouveaux canaux timbres sont bien 7, 8 et 9
FB9F	8D E1 02	STA \$02E1	Place le numéro du canal (calculé) à sa place pour SOUND
FBA2	60	RTS	Retour pour traitement par MUSIC puis par SOUND
FBA3	68	PLA	Traitement de l'erreur
FBA4	68	PLA	On dépèle une adresse de retour devenue inutile
FBA5	EE E0 02	INC \$02E0	Avant d'incrémenter pour signaler l'erreur
FBA8	60	RTS	et sortir de la routine

De FBA9 à FBCF il reste encore 39 octets pour installer les modifications de la commande PLAY.

PLAY T, B, E, D (Timbre de 0 à 15, Bruit de 0 à 15, Enveloppe, Durée)

La Commande PLAY, qui est normalement située de \$FBD0 à \$FC17, a elle aussi subi une importante adaptation. Outre sa place normale, elle « déborde » maintenant non seulement dans la fin de l'ancienne

commande SOUND, mais aussi dans l'ancienne routine « Arrondir le quotient », que nous avons squatter sans remords puisqu'elle était inopérante.

Pour bien maîtriser PLAY, il faut prendre conscience d'un important changement de philosophie dans le traitement de la commande. En effet, maintenant on ne traite plus une seule commande PLAY pour trois canaux mais deux commandes PLAY. La première est identique à la commande PLAY normale et s'adresse au PSG1 « normal ». La seconde, qui s'adresse spécifiquement au PSG0 « supplémentaire », répond à la même syntaxe, sauf qu'il faut ajouter 8 aux paramètres Timbre ET Bruit. Pour couper le son du PSG1 « normal » on utilisera donc toujours PLAY0,0,0,0 mais pour couper le

son du PSG0 « supplémentaire », il faudra utiliser PLAY8,8,0,0. Quel intérêt ? Et bien cette syntaxe permet d'avoir deux groupes séparés de trois canaux du point de vue de l'enveloppe sonore (voir notre article précédent).

Comme pour MUSIC, nous avons « vampirisé » le début de la commande PLAY pour y insérer un greffon qui va traiter des paramètres étendus. Pour ceux qui n'ont pas « L'Oric à nu » de Fabrice Broche, nous rappelons ci-dessous le début normal de cette commande.

FBD0	AD E3 02	LDA \$02E 3	Début de la commande PLAY normale
FBD3	0A	ASL	Prendre le canal timbre indiqué
FBD4	0A	ASL	Décaler 3 bits à gauche
FBD5	0A	ASL	Pour les transférer dans b5-b4-b3
FBD6	0D E1 02	ORA \$02E1	Ajouter en b2-b1-b0 les canaux « bruit »
FBD9	49 3F	EOR #\$3F	etc.

Et voici ce que ça donne après notre passage...

FBD0	20 9C C6	JSR \$C69C	Force par défaut Pb5 à 0 (adresse le PSG0 « supplémentaire »)
FBD3	AD E3 02	LDA \$02E3	Charge l'accumulateur avec le n° de canal bruit
FBD6	20 A9 FB	JSR \$FBA9	Appelle notre greffon traitement paramètres étendus pour PLAY
FBD9	49 3F	EOR #\$3F	Inchangé : c'est ici qu'on revient pour poursuivre la commande PLAY

Lequel greffon est placé dans la fin de la place libérée dans l'ancienne commande SOUND :

FBA9	48	PHA	Sauve le n° de canal bruit (valeur de 0 à 15) sur la pile
FBAA	29 08	AND #\$08	Teste le bit 3 pour savoir quel PSG est concerné
FBAC	D0 0A	BNE \$FBB8	Si le bit 3 = 1, bruit sur PSG1 « normal » demandé
FBAE	20 0B F0	JSR \$F00B	BRUIT PSG0 « supplémentaire » : Teste le Bit 3 de \$02E1 (canal Timbre)
FBB1	D0 15	BNE \$FBC8	Si le bit 3 de \$02E1 est à 1, il y a incohérence d'où erreur à signaler
FBF3	68	PLA	si b3 = 0 récupère le contenu de \$02E3 (canal bruit)
FBF4	20 FB EF	JSR \$EFFB	Combine les 3 bits de \$02E3 avec les 3 bits de \$02E1
FBF7	60	RTS	Retourne en \$FBD9 pour suite normale de PLAY (sur PSG0)
FBF8	20 0B F0	JSR \$F00B	BRUIT PSG1 « normal » : Teste le bit 3 de \$02E1 (canal Timbre)
FBFB	F0 0B	BEQ \$FBC8	Si le bit 3 de \$02E1 est à 0, il y a incohérence d'où erreur à signaler
FBFD	20 95 C6	JSR \$C695	Force Pb5 à 1 pour adresser le PSG1 « normal »
FBC0	20 02 F0	JSR \$F002	Force le bit 3 de \$02E1 à 0, pour compatibilité avec le traitement PLAY
FBC3	68	PLA	Récupère le contenu de \$02E3
FBC4	20 FB EF	JSR \$EFFB	Combine les 3 bits de \$02E3 avec les 3 bits de \$02E1
FBC7	60	RTS	Retourne en \$FBD9 pour suite normale de PLAY (sur PSG1)

Il faut noter que, dans les 2 cas, le bit 3 de \$02E3 n'a pas été remis à 0 par manque de place. Ce bit, inversé par l'EOR#\$3F en \$FBD9, va se retrouver dans le bit 6 du registre 7 de l'AY8912 et mettre ainsi le port A de ce circuit en Entrée. Ce port est utilisé pour la gestion du clavier, mais celle-ci est prévue pour forcer le port A en sortie. En conclusion, cette non remise à 0 est sans effet.

FBC8	68	PLA	TRAITEMENT DE L'ERREUR : Nettoie la pile (neutralise le PHA en \$FBA9)
FBC9	EE E0 02	INC \$02E0	Incréméte pour signaler une erreur
FBCC	68	PLA	dépille une adresse de retour devenue inutile (JSR en \$FBD6)
FBCD	68	PLA	
FBCF	60	RTS	Retourne pour poursuivre le traitement de l'erreur
FBCF	EA	NOP	Pour nettoyer le dernier octet de l'ancienne commande SOUND

Suite des routines de PLAY dans l'ancienne routine « Arrondir le quotient » (\$EFA à \$F015)

EFA	60	RTS	Retour sans rien faire pour appels à l'ancienne routine ARRONDIR QUOTIENT
EFB	0A	ASL	Routine « Combine les 3 bits de \$02E3 avec les 3 bits de \$02E1 »
EFC	0A	ASL	En entrée A = valeur de \$02E3 (canal bruit) 3 décalages arithmétiques à gauche
EFD	0A	ASL	pour placer les bit 0 à 2 en position 3 à 5
EFE	0D E1 02	ORA \$02E1	Puis OU avec \$02E1 (canal Timbre)
F01	60	RTS	
F02	AD E1 02	LDA \$02E1	Routine « Normalise \$02E1, pour compatibilité avec le traitement PLAY »
F05	29 F7	AND #\$F7	
F07	8D E1 02	STA \$02E1	Force à 0 le bit 3 de \$02E1
F0A	60	RTS	
F0B	AD E1 02	LDA \$02E1	Routine « Détermine quel PSG est appelé »
F0E	29 08	AND #\$08	Teste le bit 3 de \$02E1
F10	60	RTS	
F11	EA EA EA	NOP NOP NOP	Les 5 derniers octets de l'ancienne routine « Arrondir le quotient » sont
F14	EA EA EA	NOP NOP	remplacés par des NOP (c'est plus propre... non ?)

Ci-contre,
démonstration de l'Atmos Stéréophonique,
lors de la visu du 17 juin à Paris.



GESTION DE PB5 ET DU CLAVIER

Malgré notre expérience de PB5 (voir notre dernière mise à jour dans ce CEO-MAG), nous avons eu quelques difficultés pour venir à bout de bruits parasites et de plantage du clavier, dus aux changements de PSG. En effet, comme nous l'avons indiqué, le PSG est largement impliqué dans la gestion du clavier.

Les deux routines de permutation de PSG ont été logées dans la commande EDIT, qui n'est donc plus opérationnelle, mais bof ! C'est Dominique qui a mis le doigt sur la solution,

qui a posteriori semble pourtant évidente : après chaque changement de PSG, il faut effectuer une mise à jour en exécutant la routine « Balayer tout le clavier » de la ROM.

PB5 est mis en sortie lors de l'initialisation de l'Oric et n'est plus changé par la suite. Donc, a priori, il est inutile de spécifier que le b5 doit être en sortie.

La commande EDIT se situe de \$C692 à \$C6B2, voici le code modifié :

C692	4C E2 CA	JMP \$CAE2	Termine ici la commande EDIT en purgeant son paramètre
C695	AD 00 03	LDA \$0300	Valide le PSG1 : Place le contenu du port B dans l'accumulateur
C698	09 20	ORA #\$20	force Pb5 à 1
C69A	D0 05	BNE \$C6A1	Saut inconditionnel
C69C	AD 00 03	LDA \$0300	Valide le PSG0 : Place le contenu du port B dans l'accumulateur
C69F	29 DF	AND #\$DF	Force Pb5 à 0
C6A1	8D 00 03	STA \$0300	Remplace la nouvelle valeur sur le port B
C6A4	4C 23 F5	JMP \$F523	Balayer tout le clavier et retourner au point d'appel dans SOUND ou PLAY

AUTRES MODIFICATIONS PAR RAPPORT A LA ROM « STANDARD »

Ces modifications concernent principalement la protection de PB5 et accessoirement le copyright. Nous nous contenterons de lister les modifications effectuées.

Si besoin, reportez-vous à nos deux articles « Réalisez vos cartouches PB5 » n° 8 (mars 2000) et n°9 (présent CEO-MAG), pour avoir plus d'explications.

Correction de la routine « Configuration pour travail K7 » (de \$E76A à \$E78F)

E770	A9 70	LDA #\$70	au lieu de	LDX \$E782,Y
E772	8D 00 03	STA \$0300		LDA \$E789,Y
E775	BE 82 E7	LDX \$E782,Y		STA \$0300,X
E778	B9 89 E7	LDA \$E789,Y		DEY
E77B	9D 00 03	STA \$0300,X		BPL \$E770
E77E	88	DEY		LDA #\$40
E77F	10 F4	BPL \$E775		STA \$0300

Réduction de la taille du Copyright et ajout d'une routine corrective dans l'espace gagné.

Le Copyright est normalement situé de \$ED96 à \$EDC3.

ED96	4D 41 45 53 54 52 4F 2D 53 20 56 31 2E 32 65 0D 0A	MAETRO-S V1.2e
EDA7	60 20 44 41 43 20 32 30 30 30 0D 0A 00	© DAC 2000
EDB4	AD 00 03	LDA \$0300 Routine corrective placée dans
EDB7	29 20	AND #\$20
EDB9	09 98	ORA #\$98 la fin inutilisée
EDBB	85 99	STA \$99
EDBD	A9 0E	LDA #\$0E du copyright normal
EDBF	4C 90 F5	JMP \$F590
EDC2	EA 00	NOP BRK

Correction de la routine « Balayer une colonne du clavier » (\$F561 à \$F58F)

F562	EA EA	NOP NOP	au lieu de	LDA #\$0E
F564	20 B4 ED	JSR \$EDB4		JSR \$F590
F56E	05 99	ORA \$99		ORA #\$B8

Modification des couleurs d'encre et de papier (facilite l'identification des diverses ROM)

F914	A9 04	LDA #\$04	(encre bleue)	au lieu de	LDA #\$00	(encre noire)
F919	A9 13	LDA #\$13	(papier jaune)		LDA #\$17	(papier blanc)

Sauvez ce travail sous le nom MAEST2E.ROM.
Si tout va bien, la checksum de ce nouveau fichier doit être #1D9F (si vous êtes parti de la même ROM que nous).

Ci-contre, gros plan sur la carte mère de l'Atmos Stéréophonique, vu lors de la visu du 17 juin.

à bientôt...

